

miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University

WEIXI MA, Indiana University

DANIEL P. FRIEDMAN, Indiana University

The syntax of a programming language should reflect its semantics. When writing a `conde` expression in miniKanren, a programmer would expect all clauses share the same chance of being explored, as these clauses are written in parallel. The existing search strategy, interleaving DFS (iDFS), however, prioritize its clauses by the order how they are written down. Similarly, when a `conde` is followed by another goal conjunctively, a programmer would expect states in parallel share the same chance of being explored. Again, the answers by iDFS is different from the expectation. We have devised three new search strategies that have different level of fairness in `disj` and `conj`.

ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. miniKanren with fair search strategies. 1, 1 (May 2019), 11 pages. <https://doi.org/10.1145/nnnnnnn.nnnnnnn>

1 INTRODUCTION

miniKanren programs, especially relational interpreters, have been proven to be useful in solving many problems [1]. A subtlety in writing relational programs having large `conde` expressions, such as interpreters, is that the order of `conde` clauses can affect the speed considerably. When a `conde` expression is large enough, the left clauses consume almost all the resource and the right ones are hardly explored. The unfair `disj` of the current search strategy, interleaving DFS, is the cause. Under the hood, `conde` uses `conj` to create a goal for each clause, and `disj` to combine these goals to one. The current `disj` allocates half resource to its first goal, then allocates the other half to the rest similarly, except for the last clause which takes all the resource.

Being aware of `disj` fairness, we also investigate `conj` fairness.

We propose three new search strategies, balanced interleaving DFS (biDFS), fair DFS (fDFS), and BFS. They have different characteristics of fairness (Table. 1). We prove that the pure subset of our BFS and the BFS proposed by Seres et al [4] produce the same result when queried. But our code is shorter and runs faster. Because the two BFSs are equivalent, we do not distinguish them in most places.

2 FAIRNESS

We demonstrate the aspects (`disj` or `conj`) and levels (unfair, almost-fair, or fair) of fairness by running queries about `repeato`, a relational definition that relates a term `x` with a non-empty list whose elements are `x` (Fig. 1).

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/5-ART \$15.00

<https://doi.org/10.1145/nnnnnnn.nnnnnnn>

fairness	iDFS	biDFS	fDFS	BFS
disj	unfair	almost-fair	fair	fair
conj	unfair	unfair	unfair	fair

Table 1. fairness of search strategies

```

> (defrel (repeato x out)
  (conde
    [(== '(,x) out)]
    [(fresh (res)
      (== '(,x . ,res) out)
      (repeato x res))]))
> (run 4 q
  (repeato '* q))
'((*) (* *) (* * *) (* * * *))

```

Fig. 1. repeato and an example run

2.1 fair disj

In the following program, the three `conde` clauses differ in a trivial way. So we expect lists of each sort constitute 1/4 of the answer list. However, iDFS, the current search strategy, gives us much more lists of `a` than other sorts of lists. And some sorts (e.g. lists of `c`) are hardly found. The situation would be worse if we add more `conde` clauses.

```

;; iDFS (unfair disj)
> (run 12 q
  (conde
    [(repeato 'a q)]
    [(repeato 'b q)]
    [(repeato 'c q)]
    [(repeato 'd q)]))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))

```

We borrow the definition of *fair disj* from [4]: search strategies with *fair disj* should allocate resource evenly among disjunctive goals. Running the same program with fDFS and BFS give the following result.

```

95      ;; fDFS and BFS (fair disj)
96      > (run 12 q
97         (conde
98           [(repeato 'a q)]
99           [(repeato 'b q)]
100          [(repeato 'c q)]
101          [(repeato 'd q)]))
102      '((a) (b) (c) (d)
103        (a a) (b b) (c c) (d d)
104        (a a a) (b b b) (c c c) (d d d))
105
106
107

```

We conceive a middle place between fair and unfair – search strategies with *almost-fair disj* should allocate resource so evenly among disjunctive goals[4] that the the maximal ratio of resource is bounded by a constant. Our new search strategy, biDFS, has almost-fair *disj*. It is fair when the number of goals is a power of 2, otherwise, some goals are allocated twice more resource than the others. In the previous example, it gives the same result. And in the following example, where the *cond^e* has 5 clause, the clauses of b, c, and d are allocated more resource.

```

117      ;; biDFS (almost-fair disj)
118      > (run 16 q
119         (conde
120           [(repeato 'a q)]
121           [(repeato 'b q)]
122           [(repeato 'c q)]
123           [(repeato 'd q)]
124           [(repeato 'e q)]))
125      '((b) (c) (d) (a)
126        (b b) (c c) (d d) (e)
127        (b b b) (c c c) (d d d) (a a)
128        (b b b b) (c c c c) (d d d d) (e e))
129
130
131
132

```

2.2 fair conj

In the following program, the three *cond^e* clauses differ in a trivial way. So we expect lists of each sort constitute 1/4 of the answer list. However, search strategies with unfair *conj* (e.g. iDFS, fDFS) give us much more lists of *a* than other sorts of lists. And some sorts (e.g. lists of *c*) are hardly found. Although iDFS's *disj* is unfair in general, it is fair when there is no call to relational definition in sub-goals, including this case. The situation would be worse if we add more *cond^e* clauses. The result with biDFS, whose *conj* is also unfair, is similar, but due to its different *disj*, the position of *b* and *c* are swapped.

```

142 ;; iDFS and fDFS (unfair conj)
143 > (run 12 q
144   (fresh (x)
145     (conde
146       [(== 'a x)]
147       [(== 'b x)]
148       [(== 'c x)]
149       [(== 'd x)])
150     (repeato x q)))
151 '((a) (a a) (b) (a a a)
152   (a a a a) (b b)
153   (a a a a a) (c)
154   (a a a a a a) (b b b)
155   (a a a a a a a) (d))
156

```

Intuitively, search strategies with fair `conj` should produce each sort of lists equally frequently. Indeed, BFS does so.

```

160 ;; BFS (fair conj)
161 > (run 12 q
162   (fresh (x)
163     (conde
164       [(== 'a x)]
165       [(== 'b x)]
166       [(== 'c x)]
167       [(== 'd x)])
168     (repeato x q)))
169 '((a) (b) (c) (d)
170   (a a) (b b) (c c) (d d)
171   (a a a) (b b b) (c c c) (d d d))
172

```

A more interesting situation is when the first conjunctive goal produces infinite many states. Consider the following example, a naive specification of fair `conj` might require search strategies to produce all sorts of singleton lists, but no longer ones, which makes the strategies *incomplete*.

```

177 ;; naively fair conj
178 > (run 6 q
179   (fresh (xs)
180     (conde
181       [(repeato 'a xs)]
182       [(repeato 'b xs)])
183     (repeato xs q)))
184 '(((a)) ((b))
185   ((a a)) ((b b))
186   ((a a a)) ((b b b)))
187

```

Our solution is requiring a search strategy with *fair conj* to package states in bags, where each bag contains finite states, and to allocate resource evenly among search spaces derived from states in the same bag. The way to package depends on search strategy. And how to allocate resource among search space related to different bags is unspecified. Our definition of *fair conj* is orthogonal with completeness. For example, a naively fair strategy is fair but not complete, while BFS is fair and complete.

BFS packages states by their costs. The *cost* of a state is its depth in the search tree (i.e. the number of calls to relational definitions required to find them) [4]. In the following example, the cost of an answer is equal to the length of the inner lists plus the length of the outer list. Noted that BFS produces answers in increasing order of cost.

```
;; BFS (fair conj)
> (run 12 q
  (fresh (xs)
    (conde
      [(repeato 'a xs)]
      [(repeato 'b xs)]))
  (repeato xs q)))
'(((a)) ((b))
  ((a) (a)) ((b) (b))
  ((a a)) ((b b))
  ((a) (a) (a)) ((b) (b) (b))
  ((a a) (a a)) ((b b) (b b))
  ((a a a)) ((b b b)))
```

3 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (biDFS) has almost-fair *disj* and unfair *conj*. The implementation of biDFS differs from iDFS in the *disj* macro. We list the new *disj* with its helpers in Fig. 2. The first helper function, *split*, takes a list of goals *ls* and a procedure *k*, partitions *ls* into two sub-lists of roughly equal length, and returns the application of *k* to the two sub-lists. *disj** takes a non-empty list of goals *gs* and returns a goal. With the help of *split*, it essentially constructs a *balanced* binary tree where leaves are elements of *gs* and nodes are *disj2*, whence the name of this search strategy. In contrast, the *disj* in iDFS essentially constructs the same sort of binary tree in one of the most unbalanced forms.

4 FAIR DEPTH-FIRST SEARCH

Fair DFS (fDFS) has fair *disj* and unfair *conj*. The implementation of fDFS differs from iDFS's in *disj2* (Fig. 3). *disj2* is changed to call a new and fair version of *append-inf*. *append-inf/fair* immediately calls its helper, *append-inf/fair^*, with the first argument, *s?*, set to *#t*, which indicates that *s-inf* and *t-inf* haven't been swapped. The swapping happens at the third *cond* clause in the helper, where *s?* is updated accordingly. The first two *cond* clauses essentially copy the *cars* and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned early, is just for swapping. When the fourth and last clause runs, we know that both *s-inf* and *t-inf* are ended with a thunk. In this case, a new thunk is constructed. The new thunk calls the driver recursively. Here changing the order of *t-inf* and *s-inf* won't hurt the fairness (though it will change the order of answers). We swapped them back so that answers are produced in a more natural order.

```

236 #| [Goal] x ([Goal] x [Goal] -> Goal) -> Goal |#
237 (define (split ls k)
238   (cond
239     [(null? ls) (k '() '())]
240     [else (split (cdr ls)
241                  (lambda (l1 l2)
242                    (k (cons (car ls) l2) l1))))]))
243
244
245 #| [Goal] -> Goal |#
246 (define (disj* gs)
247   (cond
248     [(null? (cdr gs)) (car gs)]
249     [else
250      (split gs
251              (lambda (gs1 gs2)
252                (disj2 (disj* gs1)
253                       (disj* gs2))))]))
253
254
255 (define-syntax disj
256   (syntax-rules ()
257     [(disj) fail]
258     [(disj g ...) (disj* (list g ...))]))
259
260

```

Fig. 2. biDFS implementation

5 BREADTH-FIRST SEARCH

BFS is fair in both `disj` and `conj`. Our implementation is based on `fDFS` (not `iDFS`). All we have to do is apply two trivial changes to `append-map-inf`. First, rename it to `append-map-inf/fair`. Second, replace its use of `append-inf` to `append-inf/fair`.

The implementation can be improved in two aspects. First, as we mentioned in section 2.2, our BFS bag states by their cost. However, in this implementation, it is unclear where this information is recorded. Second, `append-map-inf/fair` is extravagant in memory usage. It makes $O(n + m)$ new `cons` cells every time, where n and m are the “length”s of input search spaces. We address these issues in the first subsection.

Both our BFS and Seres’s BFS [4] produce answers in increasing order of cost. So it is interesting to see if they are equivalent. We prove so in Coq. The details are in the second subsection.

5.1 optimized BFS

As we mentioned in section 2.2, our BFS bag states by their cost. The bagging information is recorded subtly – the `cars` of a search space have cost 0 (i.e. they are in the same bag), and the costs of states in `thunk` are computed recursively then increased by one. It is even more subtle that `append-map-inf/fair` and the `append-map-inf` respects the cost information. We make these facts more obvious by changing the type of search space, modifying related function definitions, and introducing a few more functions.

```

283 #| Goal x Goal -> Goal |#
284 (define (disj2 g1 g2)
285   (lambda (s)
286     (append-inf/fair (g1 s) (g2 s))))
287
288 #| Space x Space -> Space |#
289 (define (append-inf/fair s-inf t-inf)
290   (append-inf/fair^ #t s-inf t-inf))
291
292
293 #| Bool x Space x Space -> Space |#
294 (define (append-inf/fair^ s? s-inf t-inf)
295   (cond
296     ((pair? s-inf)
297      (cons (car s-inf)
298            (append-inf/fair^ s? (cdr s-inf) t-inf)))
299     ((null? s-inf) t-inf)
300     (s? (append-inf/fair^ #f t-inf s-inf))
301     (else (lambda ()
302              (append-inf/fair (t-inf) (s-inf))))))
303
304

```

Fig. 3. fDFS implementation

The new type is a pair whose `car` is a list of state (the bag), and whose `cdr` is either a `#f` or a thunk returning a search space. A falsy `cdr` means the search space is obviously finite.

Functions related to the pure subset are listed in Fig. 4 (the others in Fig. 5). We take them apart because they are compared with Seres's implementation later. The first three functions in Fig. 4 are search space constructor. `none` makes an empty search space. `unit` makes a space from one state. `step` makes a space from a thunk. The remaining functions do the same thing as before.

Luckily, the change in `append-inf/fair` also fixes the miserable space extravagance – the use of `append` helps us to reuse the first bag of `t-inf`.

Noted that some functions here constitute a *MonadPlus*: `none`, `unit`, `append-map-inf`, and `append-inf` correspond to `mzero`, `unit`, `bind`, and `mplus` respectively.

Functions implementing impure features are in Fig. 5. The first function, `elim`, takes a space `s-inf` and two continuations `ks` and `kf`. When `s-inf` contains some states, `ks` is called with the first state and the rest space. Otherwise, `kf` is called with no argument. Here 's' and 'f' means 'succeed' and 'fail' respectively. This function is an eliminator of search space, whence the name. The remaining functions do the same thing as before.

5.2 comparison with Silvija's BFS

In this section, we compare the pure subset of our optimized BFS with the BFS found in [4]. We focus on the pure subset because Silvija's system is pure. Her system represents search spaces with streams of lists of states, where each list is a bag.

```

330 (define (none) '(() . #f))
331 (define (unit s) '((,s) . #f))
332 (define (step f) '(() . ,f))
333
334 (define (append-inf/fair s-inf t-inf)
335   (cons (append (car s-inf) (car t-inf))
336         (let ([t1 (cdr s-inf)]
337               [t2 (cdr t-inf)])
338           (cond
339             [(not t1) t2]
340             [(not t2) t1]
341             [else (lambda () (append-inf/fair (t1) (t2)))]))))
342
343
344 (define (append-map-inf/fair g s-inf)
345   (foldr
346     (lambda (s t-inf)
347       (append-inf/fair (g s) t-inf))
348     (let ([f (cdr s-inf)])
349       (step (and f (lambda () (append-map-inf/fair g (f)))))
350       (car s-inf)))
351
352
353 (define (take-inf n s-inf)
354   (let loop ([n n]
355              [vs (car s-inf)])
356     (cond
357       ((and n (zero? n)) '())
358       ((pair? vs)
359        (cons (car vs)
360              (loop (and n (sub1 n)) (cdr vs))))
361       (else
362        (let ([f (cdr s-inf)])
363          (if f (take-inf n (f)) '()))))))
364
365

```

Fig. 4. new and changed functions in optimized BFS that implements pure features

To compare efficiency, we translate her Haskell code into Racket (See supplements for the translated code). The translation is fairly straightforward due to the similarity in both logic programming system and search space representation. The translated code is longer and slower. Details about efficiency difference are in section 6.

We prove the two search strategies are equivalent in Coq. Since search space can be infinite, we should use a co-inductive data type. However, Coq is too strict in the guardedness condition to accept a direct translation of the implementations. Therefore, we prove core theorems with finite search space instead. In


```

377
378 (define (elim s-inf ks kf)
379   (let ([ss (car s-inf)]
380         [f (cdr s-inf)])
381     (cond
382       [(and (null? ss) f)
383        (step (lambda () (elim (f) ks kf)))]
384       [(null? ss) (kf)]
385       [else (ks (car ss) (cons (cdr ss) f))]))))
386
387
388 (define (ifte g1 g2 g3)
389   (lambda (s)
390     (elim (g1 s)
391           (lambda (s0 s-inf)
392             (append-map-inf/fair g2
393                                   (append-inf/fair (unit s0) s-inf)))
394           (lambda () (g3 s)))))
395
396
397 (define (once g)
398   (lambda (s)
399     (elim (g s)
400           (lambda (s0 s-inf) (unit s0))
401           (lambda () (none)))))
402
403

```

Fig. 5. new and changed functions in optimized BFS that implements impure features

order to generalize the conclusion to the cases with infinite search space, we prove a few more theorems saying that whenever we query answers lower than some finite cost, we can restrict goals to truncate search spaces at some finite depth without changing the query result. (See supplements for the formal proof)

6 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 2. A hyphen means running out of memory. The first two benchmarks are taken from [2]. **reverso** is from [3]. Next two benchmarks about quine are modified from a similar test case in [1]. The modifications are made to circumvent the need for symbolic constraints (e.g. \neq , **absento**). Our version generates de Bruijnized expressions and prevent closures getting into list. The two benchmarks differ in the **cond^e** clause order of their relation interpreters. The last two benchmarks are about synthesizing expressions that evaluate to '(I love you). This benchmark is also inspired by [1]. Again, the sibling benchmarks differ in the **cond^e** clause order of their relation interpreters. The first one has elimination rules (i.e. application, **car**, and **cdr**) at the end, while the other has them at the beginning. We conjecture that iDFS would perform badly in the second case because elimination rules complicate the problem when running backward. The evaluation supports our conjecture.

benchmark	size	iDFS	biDFS	fDFS	optimized BFS	Silvija's BFS
very-recursiveo	100000	579	793	2131	1438	3617
	200000	1283	1610	3602	2803	4212
	300000	2160	2836	-	6137	-
appendo	100	31	41	42	31	68
	200	224	222	221	226	218
	300	617	634	593	631	622
reverso	10	5	3	3	38	85
	20	107	98	51	4862	5844
	30	446	442	485	123288	132159
quine-1	1	71	44	69	-	-
	2	127	142	95	-	-
	3	114	114	93	-	-
quine-2	1	147	112	56	-	-
	2	161	123	101	-	-
	3	289	189	104	-	-
'(I love you)-1	99	56	15	22	74	165
	198	53	72	55	47	74
	297	72	90	44	181	365
'(I love you)-2	99	242	61	16	66	99
	198	445	110	60	42	64
	297	476	146	49	186	322

Table 2. The results of a quantitative evaluation: running times of benchmarks in milliseconds

In general, only iDFS and biDFS constantly perform well. fDFS is just as efficient in all benchmarks but **very-recursiveo**. Both BFS have obvious overhead in many cases. Among the three variants of DFS (they all have unfair **conj**), fDFS is most resistant to clause permutation, followed by biDFS then iDFS. Among the two implementation of BFS, ours constantly performs as well or better. Interestingly, every strategies with fair **disj** suffers in **very-recursiveo** and fDFS performs well elsewhere. Therefore, this benchmark might be a special case. Fair **conj** imposes overhead constantly except in **appendo**. The reason might be that strategies with fair **conj** tend to keep more intermediate states in the memory.

7 RELATED WORKS

Edward points out a disjunct complex would be 'fair' if it is a full and balanced tree [5].

Silvija et al [4] also describe a breadth-first search strategy. We proof their BFS is equivalent to ours. However, ours looks simpler and performs better in comparison with a straightforward translation of their Haskell code.

8 CONCLUSION

We analysis the definitions of fair **disj** and fair **conj**, then propose a new definition of fair **conj**. Our definition is orthogonal with completeness.

We devise three new search strategies: balanced interleaving DFS (biDFS), fair DFS (fDFS), and BFS. biDFS has almost-fair **disj** and unfair **conj**. fDFS has fair **disj** and unfair **conj**. BFS has both fair **disj** and fair **conj**.

Our quantitative evaluation shows that biDFS and fDFS are competitive alternatives to iDFS, the current search strategy, and that BFS is less practical.

We prove our BFS is equivalent to the BFS in [4]. Our code is shorter and runs faster than a direct translation of their Haskell code.

ACKNOWLEDGMENTS

REFERENCES

- [1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017), 8.
- [2] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.
- [3] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [4] Silvija Seres, J Michael Spivey, and CAR Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
- [5] Edward Z Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue 15* (2010), 11.